

Case Study – Using iteration to animate a chart

Building on the case study *Launch an object*, this case study looks at how to animate a chart in Excel using iterative calculations. Clearly, it is difficult to show an animated chart on paper, but the three snapshots in Figure 1 should give a sense of the intent. The dotted blue line is the projected path of the object. The pink markers trace the path of the object from launch to the time of the snapshot.




Figure 1

Concept

The idea is to maintain a counter that goes from zero to some value and then resets itself to zero. In the *Launch an object* case, the path of the projectile was computed at 100 points during its flight. To leverage that work, we will use a counter that goes from 0 to 100. Depending on the value of the counter, we will show that many locations of the object. Hence, as the counter goes from 0 to 100, we will show more and more locations until the complete trajectory is traced. Then, as the counter gets to zero, the animation will reset itself.

Note that in addition to the counter cycling back to zero, we need a way to force a reset to zero at our discretion.

Implementation

Turn on iterative calculation with **Tools | Options... | Calculation** tab. In the dialog box, check on the **Iteration** checkbox and set the **Maximum Iterations** to 1.

Designate some cell, say G12 as the Reset cell. So, in G11 enter the text *Reset*. Similarly, designate F12 as the counter. In F11 enter the text *Counter*. In F12, enter the formula $=IF(OR(G12=1,F12=100),0,F12+1)$. Note that the formula, which is in F12, includes a reference to F12. That is why we need iterative calculations enabled. Each time Excel recalculates the

worksheet, it processes F12 as follows: If either G12 (the reset switch) is 1 or if the value in F12 is 100, then the value in F12 becomes zero. Otherwise, Excel adds one to the value in F12.

To see this in action, ensure that G12 (the reset switch) is empty. Then, press and let go of the F9 key a bunch of times. Each time the value in F12 will increase by 1. Once it reaches 100, it will reset to zero with the next recalculation. Hold down the F9 key and the value in F12 will repeatedly race through the cycle from 0 to 100.

To plot the in-flight trajectory, we will leverage the work already done in the previous case. Cells B13:C112 contain the horizontal and vertical coordinates of the object in flight. Each row represents one of 100 snapshots. To animate the motion of the object, as the counter goes from 0 to 100 we will show that many snapshots. In F13 enter the formula =IF(\$F\$12<=ROW()-ROW(\$F\$13),NA(),B13). In G13 enter =IF(ISNA(F13),NA(),C13). Copy F13:G13 down to rows 14:112. The result (though not all the rows up to 112) is in Figure 2.

=IF(\$F\$12<=ROW()-ROW(\$F\$13),NA(),B13)		
E	F	G
	=IF(OR(G12=1,F12=100),0,F12+1)	
	=IF(\$F\$12<=ROW()-ROW(\$F\$13),NA(),B13)	=IF(ISNA(F13),NA(),C13)
	=IF(\$F\$12<=ROW()-ROW(\$F\$13),NA(),B14)	=IF(ISNA(F14),NA(),C14)
	=IF(\$F\$12<=ROW()-ROW(\$F\$13),NA(),B15)	=IF(ISNA(F15),NA(),C15)
	=IF(\$F\$12<=ROW()-ROW(\$F\$13),NA(),B16)	=IF(ISNA(F16),NA(),C16)
	=IF(\$F\$12<=ROW()-ROW(\$F\$13),NA(),B17)	=IF(ISNA(F17),NA(),C17)
	=IF(\$F\$12<=ROW()-ROW(\$F\$13),NA(),B18)	=IF(ISNA(F18),NA(),C18)
	=IF(\$F\$12<=ROW()-ROW(\$F\$13),NA(),B19)	=IF(ISNA(F19),NA(),C19)
	=IF(\$F\$12<=ROW()-ROW(\$F\$13),NA(),B20)	=IF(ISNA(F20),NA(),C20)
	=IF(\$F\$12<=ROW()-ROW(\$F\$13),NA(),B21)	=IF(ISNA(F21),NA(),C21)
	=IF(\$F\$12<=ROW()-ROW(\$F\$13),NA(),B22)	=IF(ISNA(F22),NA(),C22)
	=IF(\$F\$12<=ROW()-ROW(\$F\$13),NA(),B23)	=IF(ISNA(F23),NA(),C23)
	=IF(\$F\$12<=ROW()-ROW(\$F\$13),NA(),B24)	=IF(ISNA(F24),NA(),C24)
	=IF(\$F\$12<=ROW()-ROW(\$F\$13),NA(),B25)	=IF(ISNA(F25),NA(),C25)

Figure 2

Create a XY Scatter chart from B12:C112. This will be the projected trajectory. Format this series with no marker and a dotted line (in the chart, double-click the plotted series, then from the Patterns tab make the necessary adjustments). Now, select and copy the range F13:G112. Select the chart, then the menu item **Edit | Paste Special...** In the resulting dialog box, specify that the data are in columns and that the first column contains X values. Format this new series with a visible marker and no line.

Depending on the value of the counter in F12, only some number of points from this second series will be shown. As the counter increases in value, more and more points will become visible creating the effect of showing the object moving across the chart.

A variation on the animation effect

As implemented above, the chart shows a trail of the object from the launch position to the current position with each point on the trail formatted the same as the current position. How about showing just the current position of the object during flight? This is very easy to implement. In the formula in F12, replace the *less than or equal to* with *not equal to* (i.e., replace \leq with \neq) and then copy F12 to F13:F112. Now, when Excel recalculates the worksheet, it will show just the current position of the object in flight as in Figure 3.




Figure 3